Listing 1. Procedury obsługi terminala dla interpretera OSI BASIC (plik OSI_BAS.S)
; STARTUP AND SERIAL I/O ROUTINES ======
; SDC_One port by gbm
MINIIO_BASE := $FE00
IO_CON      := MINIIO_BASE + 1
IO_CSR      := MINIIO_BASE + 3
IO_CSR_RXNE := 1
IO_CSR_TXE  := 2

.segment "IOHANDLER"
.org $FF00
Reset:
    LDX     #STACK_TOP
    TXS

; Display startup message
    LDY #0
ShowStartMsg:
    LDA  StartupMessage,Y
    BEQ  WaitForKeypress
    JSR  MONCOUT
    INY
    BNE  ShowStartMsg

; Wait for a cold/warm start selection
WaitForKeypress:
    JSR  MONRDKEY
    BCC  WaitForKeypress
    AND  #$DF    ; Make upper case
    CMP  #'W'    ; compare with [W]arm start
    BEQ  WarmStart
    CMP  #'C'    ; compare with [C]old start
    BNE  Reset
    JMP  COLD_START  ; BASIC cold start

WarmStart:
    JMP  RESTART  ; BASIC warm start

MONCOUT:
    PHA
SerialOutWait:
    LDA  IO_CSR
    AND  #IO_CSR_TXE
    BEQ  SerialOutWait  ; opt gbm
    PLA
    STA  IO_CON
    RTS

MONRDKEY:
    LDA  IO_CSR
    AND  #IO_CSR_RXNE
    CMP  #1
    BNE  NoDataIn	; opt gbm
    LDA  IO_CON
    SEC    ; Carry set if key available
    RTS
NoDataIn:
    CLC    ; Carry clear if no key pressed
    RTS


Listing 2. Modyfikacja testu rozmiaru pamięci (plik OSI_BAS.S)
L40D7:
    inc     LINNUM
    bne     L40DD
    inc     LINNUM+1
; SDC_One - stop memory test begore interpreter area
    Lda  LINNUM+1
    Cmp  #>BASIC_START
    Beq  MEMSIZE_DONE
L40DD:


Listing 3. Skrypt konsolidatora ld65 dla interpretera OSI BASIC
MEMORY {
    ZP:  start = $0000, size = $0100, type = rw;
    BASROM:  start = $D000, size = $2F00, fill = yes, fillval=$FF, file = %O;
    IOHANDLER:  start = $FF00, size = $FA, fill = yes, fillval=$FF, file = %O;
    VECTS:  start = $FFFA, size = $6, fill = yes, fillval=$FF, file = %O;
}

SEGMENTS {
    ZEROPAGE:  load = ZP, type = zp;
    CODE:  load = BASROM, type = ro;
    IOHANDLER:  load = IOHANDLER, type = ro;
    VECTS:  load = VECTS, type = ro;
}


Listing 4. Plik wsadowy assemble.bat
ca65 osi_bas.s -o osi_bas.o -l
ld65 -C osi_bas.cfg osi_bas.o -o osi_bas.bin
srec_cat osi_bas.bin -binary -offset 0xd000 -unfill 0xff 16 -o osi_bas.hex -intel
pause



<<<!!! LISTING 5 jest zbyt długi !!!>>>


; Listing 6. PUTSYS.ASM dla SDC_One
; Program writing memory image of CP/M system to disk.
; image from 0xE400 is copied to the disk, starting with 2nd sector
; - the 1st sector is used by bootloader
; This program is loaded to the TPA (address 0x100)
; Julia Kosowska 01'2017

#include "miniio.h"

ccp  .EQU  0e400h  ;base address of CP/M
mem  .EQU  010000h ;base address of CP/M
size  .EQU mem-ccp
nsects  .EQU	(size+127)/128 ;number of sectors to load
SPT .EQU  16
    .ORG  100H
    xra  a
    OUT  IO_MS_UNIT
    OUT  IO_MS_TRK
    Mvi  B,nsects
    mvi  C,1
    lxi  H,ccp
    lxi  D,128
next_sec:
    mov  A,C
    OUT  IO_MS_SEC
    mov  A,L
    OUT  IO_MS_A0
    mov  A, H
    OUT  IO_MS_A1
    mvi  a,IO_MS_CMD_WR
    out  IO_MS_CMD  ; write command
    DAD  D  ; next sector adress
    INR  C
    DCR  B
lop:
    JZ   fin
    mov  A,C
    CPI  SPT
    JNZ  next_sec
; advance to next track
    IN  IO_MS_TRK
    INR  A
    OUT  IO_MS_TRK
    mvi  C,0
    jmp  next_sec

fin:  ; flush buffer
    mvi	a, IO_MS_CMD_SYNC
    out	IO_MS_CMD	; write command
    jmp	$
    .end


Listing 7. GETSYS.ASM dla SDC_One
; Second stage bootloader program - read memory image of CP/M system from
; disk and jumps to BIOS cold boot routine. This program should be put into
; the first sector of the first track of the default system disk. First stage
; bootloader residing in ROM memory should load this program from disk
; to address 100h and jump to it.
; JK & gbm 01'2017
; assemble with: tasm -85 getsys.asm

#include "miniio.h"

ccp    .EQU	0e400h    ;base address of CP/M
bios   .equ	0FA00h    ;base of bios
mem    .EQU	010000h   ;end address of CP/M
size   .EQU	mem-ccp
nsects  .EQU  (size+127)/128  ;number of sectors to load
SPT  .EQU  16

    .ORG  100H
    xra  a
    OUT  IO_MS_UNIT
    OUT  IO_MS_TRK
    mvi  B, nsects
    mvi  C, 1   ; start with trk 0 sector 1
    lxi  H, ccp ; load address
    lxi  D, 128 ; sector size
next_sec:  ; load a sector
    mov  A, C
    OUT  IO_MS_SEC
    mov  A, L
    OUT  IO_MS_A0
    mov  A, H
    OUT  IO_MS_A1
    mvi  a, IO_MS_CMD_RD
    out  IO_MS_CMD  ; read
    DAD  D     ; next sector adress
    INR  C
    DCR  B
    JZ   bios  ; start the OS
    mov  A, C
    CPI  SPT
    JNZ  next_sec
[bookmark: _GoBack]               ; advance to next track
    IN   IO_MS_TRK
    INR  A
    OUT  IO_MS_TRK
    mvi  C, 0
    jmp  next_sec
    .end


Listing 8. WRBOOT.ASM dla SDC_One
; Write 128B of RAM memory starting at address 0x100 to the first
; sector of the first track of disk 0.
; JK & gbm 2017
; assemble with: tasm -85 wrboot.asm
#include "miniio.h"
    xra  A
    out  IO_MS_UNIT
    out  IO_MS_TRK
    out  IO_MS_SEC
    out  IO_MS_A0
    inr  A
    out  IO_MS_A1
    mvi  a, IO_MS_CMD_WR
    out  IO_MS_CMD  ; write
    mvi a, IO_MS_CMD_SYNC
    out IO_MS_CMD  ; synchronize
    jmp $
    .END


Listing 9. BOOT.ASM dla SDC_One
; CP/M-80 bootloader to be placed at address 0
; Read 128B of RAM memory starting at address 0x100 from the first
; sector of the first track of disk 0.
; JK & gbm 2017
; assemble with: tasm -85 boot.asm
#include "miniio.h"
    XRA  A
    OUT  IO_MS_UNIT
    OUT  IO_MS_TRK
    OUT  IO_MS_SEC
    OUT  IO_MS_A0  ; address LSB
    INR  A
    OUT  IO_MS_A1  ; address MSB
    MVI  A, IO_MS_CMD_RD
    OUT  IO_MS_CMD  ; read
    JMP  LOADER
    .org 100h
LOADER:
    .END

